# Maintainer: damir <damir@archlinux.org>
# Contributor: Robert Emil Berge <robert@rebi.no>

pkgname=libavc1394
pkgver=0.5.4
pkgrel=5
pkgdesc="A library to control A/V devices using the 1394ta AV/C commands."
arch=("x86_64")
url="https://sourceforge.net/projects/libavc1394/"
license=('LGPL')
depends=('libraw1394')
source=("https://downloads.sourceforge.net/sourceforge/libavc1394/${pkgname}-${pkgver}.tar.gz")
sha256sums=('7cb1ff09506ae911ca9860bef4af08c2403f3e131f6c913a2cbd6ddca4215b53')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr --mandir=/usr/share/man --disable-static
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
